

**Don't Let Your Annual Report Be Such a Manual Report:  
Neat New Ways from SAS® to Combine Text, Graphs and Tabular Reports  
in a Single Document**

Ben Cochran, The Bedford Group, Raleigh, NC

**ABSTRACT**

This paper looks at methods of combining all kinds of output, such as text, graphs and tables, into a single pdf file. The process is accomplished totally within SAS. This paper is the product of having done this for an organization to improve its annual report process. What originally took a few months to produce, now takes only a few minutes.

**INTRODUCTION**

This presentation is the result of a consulting project where the consultant helped the client automate its Annual Reporting process. The Client was manually generating annual reports for each of its members using Excel. They wanted to automate the process using a single solution – SAS.

The challenge – each report consisted of a body of text, the member logo, a series of plots, and a series of tables followed by more text. The client wanted to put them all together, along with their logo in a single document. Can SAS do all that?

The obvious answer to that question is 'Yes'. If the answer is 'No' then this paper would end right now. es, je

**THE REPORT COMPONENTS**

There were essentially 5 parts to this report: 1. the text file, 2. the company logo, 3. the graphical reports, 4. the tabular reports, 5. and then a report summary that originated in the same text file as part 1.




Figure 1. The Text file. .

What is NOT shown in Figure 1 is the complete file. There are a few paragraphs at the end of the file that the company wants to put at the end of the report. This file originated in a Word document, and then saved as a text file.



Figure 2. The Logo. .

The company wants to create a 'Cover page' for each of its members that would include some identifying information, including a company logo, such as the one in Figure 2.

The next report component is the graphical one. There were a series of about ten different plots that was part of the annual report.




Figure 3. Plot example.

The next report component is a series of tabular reports. There were about fifteen such tables.

## Residential - Total Kwh

Year	Average Consumers	% Change	Total kWh	% Change	Average kWh/month
2006	10,586	1.3	121,362,496	-6.8	955
2007	10,748	1.5	127,866,672	5.4	991
2008	30,271	181.7	294,125,728	1.3	810
2009	30,367	0.3	315,149,257	7.1	865
2010	30,522	0.5	346,356,995	9.9	946
2011	30,568	0.1	336,309,599	-2.9	917
2012	30,600	0.1	322,233,059	-4.2	878
2013	30,766	0.5	338,110,176	4.9	916
2014	30,976	0.7	355,240,800	5.1	956
2015	31,047	0.2	353,287,094	-0.5	948
<b>2016</b>	<b>31,193</b>	<b>0.5</b>	<b>362,273,335</b>	<b>2.5</b>	<b>968</b>
2017	31,287	0.3	367,873,364	1.5	980
2018	31,382	0.3	372,994,264	1.4	990
2019	31,499	0.4	376,870,123	0.0	997
2020	31,616	0.4	380,686,076	0.0	1,003

Figure 4. Partial Table Report.

The last part of the annual report contained summary remarks that are found at the end of the same text file used for the earlier part of the report.

Economic and Demographic Trends

The tables show important trends for both territories in the cooperative's economy. The periods represented in the charts compare projected growth rates over the study period. The economic projections come from Moody's Economy.com and provide total household data per county, total gross county product, real per capita income and total employment. Energy use projections for the residential and commercial sections are from the Energy Information Administration (EIA).

Summary

This past year has been a good one for Watts Up. The number of customers is up. Sales in megawatts are up. Profits are up as well. So, the only question left to ask Watts Up is: What's NOT up?

Figure 5. Summary remarks.

### THE SEARCH FOR A SOLUTION

The search began by looking at PROC DOCUMENT as it has the ability to store a report's individual components and then modify and replay them in a single report. The ODS DOCUMENT statement stores the actual ODS objects (or references to those objects) that are created when running a report. You can then use the DOCUMENT

procedure to rearrange, duplicate, or remove output from the results of a procedure or a database query without invoking the procedures from the original report. You can also use the DOCUMENT procedure to do the following:

1. Transform a report without rerunning an analysis or repeating a database query.
2. Modify the structure of the output.
3. Display output to any ODS output format.
4. Open and list ODS documents.
5. Combine text reports with other forms of output(like graphs, etc.) and much, much more.

With the DOCUMENT procedure, you are not limited to simply regenerating the same report. You can change the order in which objects are rendered, the table of contents, the templates that are used, macro variables, and ODS system options. The general syntax is:

```
PROC DOCUMENT <options <access-  
option(s)>>;  
  
    Optional Arguments  
  
    NAME= <libref.>member-name  
    <access-option(s)>  
        specifies the name of a new or  
        existing document and its access mode.
```

Figure 5. PROC DOCUMENT

Generating the report with PROC DOCUMENT requires a program similar to the one seen here.

```
options ls=90 nocenter nodate;
proc document name=mydoc(write);
  import textfile="&root\WUEC.txt" to ^;
run;
ods pdf file="c:\out1.pdf";
replay;
run; quit;
ods startpage = NO;
proc gplot data = WORK.sales;
  plot SALES*YEAR / VAXIS=AXIS1 HAXIS=AXIS2 href=&Myyear chref=red;
run; quit;
proc gplot data = WORK.sales2;
  plot SALES*YEAR / VAXIS=AXIS1 HAXIS=AXIS2 href=&Myyear chref=red;
run; quit;
proc report data=sales3 split="*";
  columns year avg cust prev cust pc c sales mw prev sales pc s x;
run;
ods pdf close;
```

Figure 6. The PROC DOCUMENT Solution.

The PROC DOCUMENT step reads the text file. Next, the ODS PDF statement opens the PDF destination. The REPLAY statement replays the text file in the PDF destination. That output is joined with the PROC GPLOTS and then the PROC REPORT output before the ODS PDF destination is closed.

The abridged output begins on the next page.

Executive Summary

Introduction

The Watts Up Electric Cooperative (WUEC) Power Requirements Study (PRS) provides annual projections for Consumers, seasonal peak loads, and energy sales. The study incorporates actual operating information from 2010 to the present, and forecasts ten years into the future for the existing system. The PRS is used to prepare official reports for outside agencies, commissions, and other entities. It is also used as the basis for routine studies, budgetary projections and statistical reports.

Objective

The PRS accomplishes multiple objectives. It is a systematic investigation into the member cooperative's growth history. This helps to achieve a more accurate perspective of the system's current status and a better understanding of the service area's unique features.

The Models

Monthly Models are developed for Total Purchases and Total Consumers in the System. These are correlated statistically with economic, demographic and weather variables that provide best fit models adjusted for ARI autocorrelation tendencies. Current economic and demographic variable projections are obtained from Moody's Economy.com. Blah Blah Blah...

Geography

The WUEC operates a distribution system covering the Eastern Shore and Jones Island, Florida. WUEC acquired the remaining service territory served by Pamlico River Power, and began service in January of 2008. As a result, WUEC now provides service to all of Accomackomania and Easthampton counties in Florida. The only major highway in the territory is U.S. 1, leading northward to Georgia and southward to Key West. The headquarters is located in HQville, Florida.

Economic and Demographic Trends

The tables show important trends for both territories in the cooperative's economy. The periods represented in the charts compare projected growth rates over the study period. The economic projections come from Moody's Economy.com and provide total household data per county, total gross county product, real per capita income and total employment. Energy use projections for the residential and commercial sections are from the Energy Information Administration (EIA).

Summary

This past year has been a good one for Watts Up. The number of customers is up. Sales in megawatts are up. Profits are up as well. So, the only question left to ask Watts Up is: What's NOT up?

Figure 7. The Text Portion of the Report

2.

The above report is joined with the rest of the report starting on the next page. After looking this part over, there are some desired enhancements for this part of the report. Notice the whole text file was read and placed at the beginning of the report... including the **Summary** section. Also notice the section headers like 'Executive Summary', 'Introduction', and 'Objective' have the same font, color, and size as the rest of the report. Keep these enhancements in mind as we ponder possible improvements.




Figure 8. Step 3 in the program.

Year	Average Customers	Previous Year Customers	Percent Change in Customers	Sales in MegaWatts	Previous Year Sales	Percent Change in Sales
2010	258	.	.	3,090	.	.
2011	287	258	11.5%	6,888	3,090	122.9%
2012	313	287	9.1%	13,772	6,888	99.9%
2013	348	313	11.3%	6,965	13,772	(49.4%)
2014	382	348	9.7%	3,056	6,965	(56.1%)
2015	399	382	4.5%	10,116	3,056	231.0%
2016	481	399	20.4%	10,234	10,116	1.2%
2017	541	481	12.6%	12,390	10,234	21.1%
2018	603	541	11.5%	15,445	12,390	24.7%
2019	648	603	7.5%	15,055	15,445	(2.5%)

Figure 9. Partial PROC REPORT output.



After viewing the report, it is clear that PROC DOCUMENT did, in fact, join the text with graphs and tables into a single document. However, the following enhancements were placed on the 'wish list':

1. A cover sheet with the company logo.
2. Split the text file so that some of appears at the beginning of the report, and some appears at the end.
3. Use different font sizes and colors in the text portions of the report.

Can we make these enhancements with PROC DOCUMENT? If not, is there some other way to accomplish the wish list enhancements in SAS? If the answer to this last question is no, then we have come to the end of this paper. Well, we have not come to the end of the paper yet, so there must be some way to accomplish our goals.

## THE COVE SHEET

Lets take a quick look at the cover sheet.




Figure 10. The Cover Sheet.

Notice the embedded logo as well as the different font sizes and colors. What neat, new tool from SAS allows us to create cover sheets like this? Was it PROC ODSTEXT, or some other new ODS tool? Was it the Report Writing Interface (RWI)? Let's look at the program to see how the cover sheet was created.

```

%let mabr=WUEC;
title;
footnote;
options nodate nonumber ;
ods escapechar='^';
ods pdf file='c:\ben\test.pdf';
title3 j=c "^S={preimage='&root\&mabr..GIF?height=2.00in&width=5.5in'}";
title8 ' ';
data _null_;
  file print n=ps ls=145 ll=remain;
  put #8 @1 "^S={font_size=44pt color=purple font=(Arial) } Annual Report ";
  put / @2 "^S={font_size=16pt color=blue} Report Date: &sysdate9";
  put / @2 "^S={font_size=16pt color=blue} Prepared by Executive Council";
  put / @2 "^S={font_size=16pt color=blue} Prepared at Corporate Head Quarters";
run;
ods pdf close;

```

Figure 11. The Cover Sheet program.

The neat new tool from SAS that allows us the power and flexibility to create the cover sheet is the DATA \_NULL\_ step. Actually, DATA \_NULL\_ is not new, but it has been given some relatively new tools and power in conjunction with ODS. Specifically, it's the ability to do 'in-line-formatting'.

Notice the ODS ESCAPECHAR statement. It specifies the special character that is the in-line formatting symbol. In-line formatting enables you to change styles in titles, footnotes and text strings. In-line formatting allows us to place the company logo (GIF file) in the TITLE3 statement and then control its height and width.

## THE FIRST TEXT SECTION

The text document needs to be divided into two sections, one that goes at the front of the report, and one that goes at the end. The **Introduction, Objective, The Models and Geography** sections go before the graphs and tables. Then the text starting with **Economic and Demographic Trends** goes at the end of the report. We need to find a way to read the text file until we see the line that says "**Economic and Demographic Trend**". How can we do this in SAS?

```

territory is U.S. 1, leading northward to Georgia and southward to Key West. The
headquarters is located in HQville, Florida.

Economic and Demographic Trends

The tables show important trends for both territories in the cooperative's economy. The
periods represented in the charts compare projected growth rates over the study period.
The economic projections come from Moody's Economy.com and provide total household data
per county, total gross county product, real per capita income and total employment.
Energy use projections for the residential and commercial sections are from the Energy
Information Administration (EIA).

Summary

This past year has been a good one for Watts Up. The number of customers is up. Sales
in megawatts are up. Profits are up as well. So, the only question left to ask Watts Up
is: What's NOT up?

```

Figure 12. Partial Text file

Not only do we want to divide the text file, we also want to make "Executive Summary" have a bigger font size and color. We also want to make the section dividers have a different font size and color also. See the next figure.




Figure 13. The First Text section of the Report.

The following DATA \_NULL\_ step gives us exactly what we want.

1. The first section (see the bracket) reads a row until it encounters the row that says "Executive Summary". There may be other text above this line that we don't want in the report.
2. Notice the two INPUT statements. The first one is used to determine the record length, then the second one reads the entire row into a variable named LINE. This method is commonly used to read records with varying lengths.
3. Once we have read the line that says "Executive Summary, we fall out of the loop and write those words with dark blue letters and a font size of 13pt using in-line formatting.

```

data _null_;
  infile "&root\&mbr..txt" pad lrecl=135 length=linelen ;
  length line $ 135.;
  if _n_=1 then do while(line not = 'Executive Summary ');
    1. input @; lnln=linelen;
    input @1 line $varying135. lnln; 2.
  end;
  file print ls=136 linesleft=remain ; ll=remain ;
  if _n_ = 1 then
    3. put @1 "^S={font_weight=Bold color=darkblue font_size=13pt just=left}"
      line;

```

Figure 14. Partial DATA Step.

The next part of the DATA step:

1. Stops when the line containing 'Economic and Demographic Trends' is encountered.
2. Writes the section dividers in blue with a font size of 11.25 pt. using in-line formatting.

```

else if _n_ gt 1 then do;
  input @; lnln=linelen;
  input @1 line $varying135. linelen;
  1. if line = 'Economic and Demographic Trends ' then stop;
  2. if line in('Introduction ', 'Objective ', 'The Models ',
    'Geography ') and line_len lt 67 then do;
    put; * writes a blank line;
    put @1 "^S={font_weight=Bold color=blue
      font_size=11.25pt just=left}" line;
  end;

```

Figure 15. Partial DATA Step. .

The only thing left to do while processing the text file is to simply read a line and then write it out. Notice the PUT statement below.

```

else do;
  put "^S={color=black font_size=9.75pt just=left}" line;
end;
if remain le 18 then put _page_ ;
end;
run;

```

Figure 16. The Last Part of the DATA Step. .

Notice the **PUT\_PAGE\_** statement. This DATA step has a way of knowing when its getting close to the end of the page using the variable **REMAIN**. **REMAIN** is created with the **LINESLEFT** option on the **FILE** statement. It keeps track of how many lines are left in the output page buffer. When the number of lines left are less than or equal to 18, this DATA step starts writing to a new page. This concludes the creation of the text portion at the beginning of the report.

The program that processes the text for the end of the report is shown next.

```
data post_page;
  infile "&root\&mbr..txt" pad lrecl=135 length=linelen ;
  length line $ 135.;
  1. if _n_ =1 then do while(line not = 'Economic and Demographic Trends ');
      input @; lnln=linelen;
      input @1 line $varying135. lnln;
  end;
  file print ls=135 linesleft=remain; ll=remain;

  2. if _n_ = 1 then put @1 "^S={font_weight=Bold color=blue
                          font_size=12.0pt just=left}" line;
```

Figure 17. The First Part of the Last DATA Step.

Notice the following:

1. The DO WHILE loop forces this DATA step to read rows from the text file and basically throw them away until it reaches the row that contains "Economic and Demographic Trends".
2. Then, it falls out of the loop and writes these words in blue with a font size of 12.0pt using in-line formatting.

The last part of the DATA step writes section dividers( like "Summary") in blue and regular lines (non section dividers) in black. .

```
else if _n_ gt 1 then do;
  input @; lnln=linelen;
  input @1 line $varying140. lnln;
  if line in('Summary ') then /* or any other section dividers */
    put / @1 "^S={font_weight=Bold color=blue
              font_size=12.0pt just=left}" line;
  else put @1 "^S={font_size=9.75pt color=black just=left} " line;

  if ll le 20 then put / _page_;
end;
run;
```

Figure 18. The Last Part of the Last DATA Step.

This concludes the processing of the text file. The results of this DATA step are shown at the top of the next page.

## Economic and Demographic Trends

The tables show important trends for both territories in the cooperative's economy. The periods represented in the charts compare projected growth rates over the study period. The economic projections come from Moody's Economy.com and provide total household data per county, total gross county product, real per capita income and total employment. Energy use projections for the residential and commercial sections are from the Energy Information Administration (EIA).

### Summary

This past year has been a good one for Watts Up. The number of customers are up. The sales in dollars are up. Profits are also up. Everything looks positive. So, the only question left for WUEC is: What's NOT Up?

Figure 19. The Report Summary

## PUTTING THE REPORT TOGETHER

All three DATA steps are put into separate Macro programs.

1. The cover sheet DATA step is put into a macro program called **%CoverSheet**.
2. The first text section DATA step is put into a macro program called **%Section1**.
3. The last text section DATA step is put into a macro program called **%Last**.
4. The PROC REPORT step is put into a macro program called **%procreport**.

The final part of the program is shown here...

```
*-----*;  
* Putting the Report Together. *;  
*-----*;  
ods pdf file='c:\ben\PRS.pdf';  
  %CoverSheet;  
  ods startpage=Yes;  
  title;  
  %Section1;  
  ods startpage=Yes;  
PROC GPLOT DATA = WORK.sales;  
  PLOT SALES*YEAR / VAXIS=AXIS1 HAXIS=AXIS2 href=&Myyear chref=red;  
RUN; QUIT;  
ods startpage=Yes;  
PROC GPLOT DATA = WORK.sales2;  
  PLOT SALES*YEAR / VAXIS=AXIS1 HAXIS=AXIS2 href=&Myyear chref=red;  
RUN; QUIT;  
ods startpage=Yes;  
  %procreport;  
  ods startpage=Yes;  
  %last;  
ods pdf close;
```

Figure 20. The Final Part of the Program.

## **CONCLUSION**

Using some of the report writing tools provided by SAS enables the user to create all kinds of reports. Here, the author used some new and old tools in order to create the kind of reports needed by the end user group. One of the best tool sets for creating SAS reports has been the ODS (Output Delivery System). This body of report writing tools and capability is very powerful and continues to grow. However, in the end, the tool needed to give the client their desired report was our old friend, DATA \_NULL\_. When you combine some of the features of ODS with the DATA step, you have a very powerful tool indeed.

## **ACKNOWLEDGMENTS**

I would like to thank my clients for this past year in posing this report writing challenge to me. Their challenges have been the source of inspiration for this presentation. And, as always, I want to thank the kind and patient folks in the Technical Support division at SAS. Without their help, this paper would not have been possible.

## **CONTACT INFORMATION**

If you have any questions or comments, the author can be reached at:

Ben Cochran  
The Bedford Group  
3224 Bedford Avenue  
Raleigh, NC 27607  
Work Phone: 919.741.0370  
Email: bencochran@nc.rr.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.